Solución problema Nº 13, Prácticas 2000-2001

Jose Mª Goicolea, octubre 2000

Un oscilador armónico está formado por una partícula con masa, un muelle y un amortiguador, y queremos caracterizar sus parámetros mediante una serie de ensayos. Se ha observado en éstos que:

- El tiempo de relajación es $\tau = .5$ s (se entiende por tiempo de relajación aquel al cabo del cual la amplitud de oscilación se reduce en un factor 1/e);
- La frecuencia de resonancia cuando se aplica en la base del muelle un movimiento impuesto armónico es $\Omega_r = 16.7$ rad/s.
- La frecuencia de resonancia es muy similar a la de oscilación propia

Se pide calcular la frecuencia propia de oscilación y la tasa de amortiguamiento crítico del oscilador.

> **restart:**

Ecuación que establece la propiedad del tiempo de relajación:

$$ecu1 := \frac{e^{(-\xi\, \omega_0\, \tau)}}{e^0} = \frac{1}{e}$$

>

Despejando el valor de ξ , tasa de amortiguamiento crítico:

> **solu1:=solve(ecu1,xi);**

> **ec1:=xi=solu1;**

$$solu1 := \frac{1}{\omega_0\, \tau}$$

$$ec1 := \xi = \frac{1}{\omega_0\, \tau}$$

La amplitud de la oscilación en régimen permanente (obsérvese que, al tratarse de una excitación por movimiento de la base, el numerador depende también de la frecuencia de la excitación Ω , al contrario del caso en que la excitación sea directamente una fuerza armónica):

$$B(\Omega) := \frac{A\, \Omega^2}{\sqrt{(\omega_0^2 - \Omega^2)^2 + 4\, \xi^2\, \omega_0^2\, \Omega^2}}$$

>

Resolución del máximo para calcular la frecuencia de resonancia:

> **ecu2:=0=diff(B(Omega),Omega);**

$$ecu2 := 0 = 2\,\frac{A\,\Omega}{\sqrt{\omega_0^4 - 2\,\omega_0^2\,\Omega^2 + \Omega^4 + 4\,\xi^2\,\omega_0^2\,\Omega^2}} - \frac{1}{2}\,\frac{A\,\Omega^2\,(-4\,\omega_0^2\,\Omega + 4\,\Omega^3 + 8\,\xi^2\,\omega_0^2\,\Omega)}{(\omega_0^4 - 2\,\omega_0^2\,\Omega^2 + \Omega^4 + 4\,\xi^2\,\omega_0^2\,\Omega^2)^{\left(\frac{3}{2}\right)}}$$

> **solu2:=solve(ecu2,Omega);**

$$solu2 := 0, \frac{\sqrt{1 - 2\,\xi^2}\,\omega_0}{-1 + 2\,\xi^2}, -\frac{\sqrt{1 - 2\,\xi^2}\,\omega_0}{-1 + 2\,\xi^2}$$

De las anteriores la solución válida es la tercera (positiva y distinta de 0):

> **ec2:=Omega[r]=solu2[3];**

$$ec2 := \Omega_r = -\frac{\sqrt{1 - 2\,\xi^2}\,\omega_0}{-1 + 2\,\xi^2}$$

Sustituimos en ella el valor de ξ antes hallado y resulta una única ecuacion para ω_0 :

> **ec12:=subs(ec1,ec2);**

$$ec12 := \Omega_r = -\frac{\sqrt{1 - \frac{2}{\omega_0^2\,\tau^2}}\,\omega_0}{-1 + \frac{2}{\omega_0^2\,\tau^2}}$$

Resolviendo se obtienen las soluciones para ω_0 buscadas:

> **solu12:=solve(ec12,omega[0]);**

$$solu12 := \frac{1}{2}\,\frac{\sqrt{2}\,\sqrt{\Omega_r^2\,\tau^2 + \sqrt{\Omega_r^4\,\tau^4 - 8\,\Omega_r^2\,\tau^2}}}{\tau}, -\frac{1}{2}\,\frac{\sqrt{2}\,\sqrt{\Omega_r^2\,\tau^2 + \sqrt{\Omega_r^4\,\tau^4 - 8\,\Omega_r^2\,\tau^2}}}{\tau},$$
$$\frac{1}{2}\,\frac{\sqrt{2}\,\sqrt{\Omega_r^2\,\tau^2 - \sqrt{\Omega_r^4\,\tau^4 - 8\,\Omega_r^2\,\tau^2}}}{\tau}, -\frac{1}{2}\,\frac{\sqrt{2}\,\sqrt{\Omega_r^2\,\tau^2 - \sqrt{\Omega_r^4\,\tau^4 - 8\,\Omega_r^2\,\tau^2}}}{\tau}$$

Para ver cuál de estas 4 soluciones es la válida, sustituimos los datos numéricos del enunciado

```
> datos:={Omega[r]=16.7*(2*Pi),tau=1/2}:
evalf(datos);
```

$$\{\Omega_r = 104.9291946, \tau = .5000000000\}$$

```
> solu12f:=evalf(subs(datos,[solu12]));
```

$$solu12f := [104.8910390, -104.8910390, 2.829455778, -2.829455778]$$

La solución "buena" es la 1, que es positiva y próxima a la frecuencia natural. Con ello se calcula la respuesta pedida:

```
> soluf:=omega[0]=solu12f[1];
```

```
> subs(soluf,datos,ec1);
```

$$soluf := \omega_0 = 104.8910390$$

$$\xi = .01906740575$$

Solución "a mano".

Desarrollando la solucion de las ecuaciones a mano, resultan dos soluciones para la ecuación bicuadrática, representando los cuadrados de ω_0 :

```
> x1:=(Omega[r]^2+Omega[r]*sqrt(Omega[r]^2-8/tau^2))/2;
```

```
> x2:=(Omega[r]^2-Omega[r]*sqrt(Omega[r]^2-8/tau^2))/2;
```

$$x1 := \frac{1}{2}\Omega_r^2 + \frac{1}{2}\Omega_r\sqrt{\Omega_r^2 - \frac{8}{\tau^2}}$$

$$x2 := \frac{1}{2}\Omega_r^2 - \frac{1}{2}\Omega_r\sqrt{\Omega_r^2 - \frac{8}{\tau^2}}$$

Las soluciones respectivas son:

```
> sol[1]:=evalf(sqrt(subs(datos,x1)));
```

```
> sol[2]:=evalf(sqrt(subs(datos,x2)));
```

$$sol_1 := 104.8910390$$

$$sol_2 := 2.829455955$$

La solución buena es la primera,

```
> subs(omega[0]=sol[1],datos,xi=1/(tau*omega[0]));
```

$$\xi = .01906740575$$